APPLICATION OF NASTRAN TO LARGE DEFLECTION SUPERSONIC FLUTTER OF PANELS

by

Chuh Mei
Vought Corporation, Hampton, Virginia

and

James L. Rogers, Jr.
NASA-Langley Research Center, Hampton, Virginia

SUMMARY

Flat panel flutter at high supersonic Mach number is analyzed using NASTRAN Level 16.0 by means of modifications to the code. Two-dimensional plate theory and quasi-steady aerodynamic theory are employed. The finite element formulation and solution procedure are presented. Modifications to the NASTRAN code are discussed. Convergence characteristics of the iteration processes are also briefly discussed. Effects of aerodynamic damping, boundary support condition and applied in-plane loading are included. Comparison of nonlinear vibration and linear flutter results with analytical solutions demonstrate that excellent accuracy is obtained with NASTRAN.

INTRODUCTION

Panel flutter is the self-excited oscillation of the external skin of a flight vehicle when exposed to an airflow along its surface. The classic approach using linear structural theory indicates that there is a critical (or flutter) dynamic pressure above which the panel motion becomes unstable. Since the linear theory does not account for structural nonlinearities, it can only determine the flutter boundary and can give no information about the flutter oscillation itself. A great quantity of literature exists on linear panel flutter (e.g. refs. 1 and 2 plus others too numerous to mention).

For large deflections, the nonlinear effects, mainly due to midplane stretching forces, restrain the panel motion to bounded limit cycle oscillations with increasing amplitude as dynamic pressure increases. Therefore, for realistic assessments and understanding of panel flutter, the nonlinear theory should be used. An excellent survey on both linear and nonlinear panel flutter through 1970 is given by Dowell (ref. 3).

To investigate large amplitude panel flutter, a number of approaches can be used. A modal approach with direct numerical integration has been used by Dowell (refs. 4 and 5). The major disadvantage in using this approach is its long computing time. The harmonic balance method can be used to determine limit cycles; see for example, Eastep and McIntosh (ref. 6) and Kuo et. al. (ref. 7). This approach, however, is quite complicated in mathematic manipulations. Morino (refs. 7 and 8) also used the pertubation method to obtain neighboring solutions to the linear problem.

The finite element method has been used successfully in investigating linear panel flutter (refs. 9 to 15). Because of its versatile applicability, effects of aerodynamic damping, complex panel configuration (e.g. delta planform in ref. 11, and rhombic planform in ref. 13), flow angularity, midplane forces, and anisotropic material properties can be conveniently included. Recently, the finite element method has been applied successfully in large amplitude vibrations of beam and plate structures (refs. 16 to 18). Thus, it is logical to extend the finite element application to study the limit cycle oscillations of panels.

The purpose of this paper is to describe a large deflection supersonic panel flutter capability available for NASTRAN Level 16.0 by means of DMAP sequences and modifications of the code. The paper includes a brief discussion of the theoretical formulation and solution procedure. Effects of aerodynamic damping, initial in-plane loading and boundary support condition are included. DMAP sequences required for nonlinear panel flutter analysis and an example of input bulk data are given in the Appendices.

SYMBOLS

a	length
$[a]$	nonsymmetric aerodynamic matrix
$c = (w)_{max}$	amplitude of oscillation
$D = \dfrac{Eh^3}{12(1-\nu^2)}$	bending rigidity
$[d]$	aerodynamic damping matrix
E	modulus of elasticity
$\{f\}$	interpolation function

g_A	aerodynamic damping parameter, equation (21)
h	thickness
$i = \sqrt{-1}$	
$[k]$	stiffness matrix
$[k^d]$	differential stiffness matrix
$[k^g]$	geometrical stiffness matrix
M	Mach number
$[m]$	mass matrix
N_x	inplane force due to deflection, tension positive
N_{xo}	applied inplane force, tension positive
p	aerodynamic force
$\{Q\}$	generalized aerodynamic force
q	dynamic pressure
t	time
$\{u\}$	nodal displacements
V	flow velocity
w	deflection
x,y,z	coordinates
α	damping factor
$\beta = \sqrt{M^2 - 1}$	
ε	norm
κ	complex eigenvalue, equation (19)
$\lambda = \frac{2qa^3}{\beta D}$	dynamic pressure parameter

μ	aerodynamic damping coefficient, equation (8)
ν	Poisson's ratio
ρ	panel mass density
ρ_A	air mass density
$\{\phi\}$	eigenvector
$\Omega = \alpha + i\,\omega$	response of system
ω	frequency
ω_o	reference frequency

Subscripts:

aa	analysis
ee	element

THEORETICAL FORMULATION AND ITS SOLUTION

Formulation of Matrix Equation of Motion

The panel is represented by a flat thin plate of unit width in bending as shown in figure 1. The transverse dynamic equilibrium equation may be written as:

$$D \frac{\partial^4 w}{\partial x^4} - (N_x + N_{xo}) \frac{\partial^2 w}{\partial x^2} + \rho h \frac{\partial^2 w}{\partial t^2} = p \qquad (1)$$

Where

$$N_x = \frac{Eh}{2a} \int_o^a \left(\frac{\partial w}{\partial x} \right)^2 dx \qquad (2)$$

is the membrane force induced by large deflections, and N_{xo} is the initial in-plane loading. For sufficiently high supersonic speeds ($M > 1.6$), the aerodynamic pressure can be described by the two dimensional aerodynamic theory:

$$p(x,y,t) = -\frac{2q}{\beta}\left[\frac{\partial w}{\partial x} + \frac{1}{V}\,\frac{M^2-2}{M^2-1}\,\frac{\partial w}{\partial t}\right] \tag{3}$$

In the finite element method, the stiffness equations of motion for a plate element under the influence of elastic, initial in-plane, large deflection, and inertia forces (ref. 17) with the inclusion of aerodynamic forces may be written as:

$$([k_{ee}] + [k^d_{ee}] + [k^g_{ee}])\{u_e\} + [m_{ee}]\{\ddot{u}_e\} = \{Q(t)\} \tag{4}$$

The stiffness $[k_{ee}]$, differential stiffness $[k^d_{ee}]$, and mass $[m_{ee}]$ matrices have been well developed for almost every plate finite element available. The geometrical stiffness matrix $[k^g_{ee}]$ has been derived in references 16 and 17 for beam and rectangular plate elements. The development of the aerodynamic matrices follows the method proposed by Olson (refs. 9 and 11). The virtual work, U, of the aerodynamic force is

$$\begin{aligned} U &= Q_j u_j \\ &= \iint p(x,y,t)\ w\ dx\ dy \end{aligned} \tag{5}$$

Assuming the displacements are exponential functions of time

$$w(x,t) = w(x)e^{\Omega t} \tag{6}$$

where, in general, $\Omega = \alpha + i\omega$. Substituting expressions for the aerodynamic pressure, equation (3), and the displacement functions, equation (6), the virtual work becomes

$$U = \left\{ -\frac{2q}{\beta} \int \frac{dw}{dx} w \, dx \; - \mu \Omega \int w^2 \, dx \right\} e^{\Omega t} \tag{7}$$

where $\mu = \dfrac{2q}{V} \; \dfrac{M^2-2}{\beta^3}$ (8)

The deflection function for a particular element is usually assumed in the form:

$$w(x) = \sum_j f_j \; (x) \; u_j = \{f\}^T \{u\} \tag{9}$$

where f_j is the interpolation function corresponding to the element nodal degree-of-freedom u_j. Introducing the expression for w(x), equation (7) yields

$$U = \left\{ -\frac{2q}{\beta} \{u_e\}^T [a_{ee}] \{u_e\} - \mu \Omega \; \{u_e\}^T [d_{ee}] \{u_e\} \right\} e^{\Omega t} \tag{10}$$

where

$$[a_{ee}] = \int \left\{ \frac{\partial f}{\partial x} \right\} \{f\} \, dx \tag{11}$$

is the non-symmetric aerodynamic matrix and

$$[d_{ee}] = \int w^2 \, dx \tag{12}$$

is the aerodynamic damping matrix. The generalized aerodynamic forces are

$$Q_j = \frac{\partial U}{\partial u_j} = -\left(\frac{2q}{\beta} [a_{ee}] + \mu\Omega \ [d_{ee}]\right)\{u_e\} e^{\Omega t} \tag{13}$$

and their substitution into equation (4) yields the dynamic equilibrium of the panel in the form:

$$\begin{aligned}([k_{ee}] + [k^d_{ee}] + [k^g_{ee}] + \frac{2q}{\beta} [a_{ee}] \\ + \Omega^2 [m_{ee}] + \mu\Omega [d_{ee}]) \{u_e\} = 0\end{aligned} \tag{14}$$

The aerodynamic damping matrix, equation (12), can be related to the mass matrix by the expression:

$$[d_{ee}] = \frac{1}{\rho h} \ [m_{ee}] \tag{15}$$

and equation (14) takes the final form for a finite element as,

$$\begin{aligned}([k_{ee}] + [k^d_{ee}] + [k^g_{ee}] + \frac{2q}{\beta} [a_{ee}] \\ + \Omega^2 [m_{ee}] + \frac{\mu\Omega}{\rho h} [m_{ee}]) \{u_e\} = 0\end{aligned} \tag{16}$$

Solution Procedure

Assembling the finite elements, applying the kinematic boundary conditions, and dividing by $(\frac{D}{a^3})$ equation (16) leads to a nondimensional eigenvalue problem of the form:

$$([k_{aa}] + [k^d_{aa}] + [k^g_{aa}] + \lambda\ [a_{aa}] - \kappa\ [m_{aa}])\ \{u_a\} = 0 \qquad (17)$$

where

$$\lambda = \frac{2qa^3}{\beta D} \qquad (18)$$

and

$$\kappa = -\frac{\rho h a^4}{D}\ \Omega^2 - \lambda\ \frac{M^2 - 2}{\beta^2}\ \frac{a}{V}\ \Omega \qquad (19)$$

are the nondimensional dynamic pressure parameter and eigenvalues, respectively. The eigenvalues can be put into more convenient form as

$$\kappa = -\frac{\Omega^2}{\omega_o{}^2} - g_A\ \frac{\Omega}{\omega_o} \qquad (20)$$

where

$$g_A = \frac{M^2 - 2}{\beta^3}\ \frac{\rho_A V}{\rho h \omega_o} \qquad (21)$$

is the nondimensional aerodynamic damping parameter, and

$$\omega_o = \sqrt{\frac{D}{\rho h a^4}} \qquad (22)$$

is a convenient frequency scale. For typical panels, g_A ranges from 0 to 50 approximately, as given in figure 2 of reference 2.

In determining the eigenvalues κ in equation (17) for a given dynamic pressure λ, the iterative procedure and equivalent linearization technique discussed in detail in reference 17 was employed. A simple flow diagram of the procedure is shown in figure 2. The solution procedure is illustrated briefly as follows. For a given λ, first the linear flutter problem is solved

$$\kappa[m_{aa}] \{\phi\}_0 = ([k_{aa}] + [k^d_{aa}] + \lambda [a_{aa}]) \{\phi\}_0 \tag{23}$$

where $\{\phi\}_0$ represents the linear mode shape normalized by its maximum components. The first approximate displacement is then expressed in the form

$$\{u_a\}_1 = c \, \text{Real}(\{\phi\}_0 \, e^{(\alpha + i\omega)t}) \tag{24}$$

where c is a given amplitude of panel oscillations, and α and ω are the panel response parameters related to κ and g_A by equation (30). An equivalent geometrical stiffness matrix $[k^g_{aa}]_{eq}$ now can be obtained using $\{u_a\}_1$, and equation (17) is approximated by a linearized eigenvalue equation of the form

$$\kappa[m_{aa}] \{\phi\}_1 = ([k_{aa}] + [k^d_{aa}] + [k^g_{aa}]_{eq} + \lambda [a_{aa}]) \{\phi\}_1 \tag{25}$$

where κ is the eigenvalue associated with amplitude c, and $\{\phi\}_1$ is the corresponding mode shape. The iterative process can be repeated until a convergence criterion is satisfied as shown in figure 2. The maximum displacement norm convergence criterion proposed in reference 19 was used in the present study and is defined as

$$\|\varepsilon\|_u = \max_j \left| \frac{\Delta u_j}{u_{j,ref}} \right| \tag{26}$$

where Δu_j is the change in displacement component j during iteration cycle n, and $u_{j,ref}$ is the reference displacement. The reference displacement is the largest displacement component of the corresponding "type". For instance in a panel flutter problem involving deflections and rotations, the reference displacement is the largest deflection component and the largest rotation, respectively. In addition a frequency norm is also introduced in the present study and is defined as

$$\|\varepsilon\|_f = \left|\frac{\Delta\kappa_n}{\kappa_n}\right| \tag{27}$$

where $\Delta\kappa_n$ is the change in eigenvalue during iteration cycle n.

A typical plot of the maximum and frequency norms versus number of iterations for a simply supported panel is shown in figure 3. A modified absolute norm and a modified Euclidean norm defined in reference 19 were also calculated. They fall in between the maximum and frequency norms, and therefore, are not plotted on the figure. In the examples presented in the following section, convergence is considered achieved whenever any one of the norms reaches a value of 10^{-3}.

Equation (17) indicates that when $\lambda=0$ the problem degenerates into large amplitude vibrations of invacuo panels. The matrices [k], $[k^d]$, $[k^g]$, and [m] are all symmetric and the eigenvalues are real and positive. As λ is increased from zero, two of these eigenvalues will usually approach each other and coalesce to κ_{cr} at $\lambda = \lambda_{cr}$, and become complex conjugate pairs

$$\kappa = \kappa_R \pm i\,\kappa_I \tag{28}$$

for $\lambda > \lambda_{cr}$. Here λ_{cr} is considered to be the lowest value of λ for which coalescence occurs among all limit cycle amplitudes and usually corresponds to c = 0. A typical plot of κ versus λ is shown in figure 4. In the absence of aerodynamic damping ($g_A = 0$), the flutter boundary simply corresponds to λ_{cr}. When λ is below λ_{cr}, any disturbance to the panel decays and $(c/h) \to 0$.

For $\lambda > \lambda_{cr}$, a periodic limit cycle oscillation exists which increases in amplitude as λ increases. This can be seen more clearly by noting that the eigenvalue with a negative imaginary part leads to an instability (see ref. 13) and relating the complex eigenvalues to the panel response parameters α and ω as follows. Rewrite equations (20) and (28) as

$$\left(\frac{\Omega}{\omega_o}\right)^2 + g_A \frac{\Omega}{\omega_o} + (\kappa_R - i\,\kappa_I) = 0 \qquad (29)$$

which can be solved for Ω to give

$$\frac{\Omega}{\omega_o} = \frac{\alpha}{\omega_o} + i\,\frac{\omega}{\omega_o}$$

$$= \left(-\frac{g_A}{2} + \psi\right) + i\left(\frac{\kappa_I}{2\psi}\right) \qquad (30)$$

where

$$\psi = \pm\frac{1}{\sqrt{2}}\left\{\sqrt{\left[\left(\frac{g_A}{2}\right)^2 - \kappa_R\right]^2 + \kappa_I^{\,2}} + \left[\left(\frac{g_A}{2}\right)^2 - \kappa_R\right]\right\}^{1/2} \qquad (31)$$

The complete panel behavior is characterized by plotting the variation of $\alpha + i\omega$ with increasing dynamic pressure λ. Amplitude increases when α becomes positive. A typical plot is shown in figure 5.

MODIFICATIONS TO THE NASTRAN CODE

To incorporate this new capability into NASTRAN, four existing NASTRAN subroutines must be modified. These subroutines are DBAR, KBAR, SDR1A (SDR1AZZ on CDC computers because of multiple entry points), and XMPLBD. DBAR was modified in the same way as shown in reference 17.

Subroutine KBAR was modified to calculate the aerodynamic matrix $[a_{ee}]$. This matrix is multiplied by the parameter DPMN = $2q/\beta$. DPMN is input via a PARAM card in the BULK DATA deck. DPMN is passed to KBAR through blank common from module EMG. The new EMG calling sequence allowing for the DPMN parameter is shown as follows:

```
EMG     EST,CSTM,MPT,DIT,GEOM2,/KELM,KDICT,MELM,MDICT,,/V,N,NOKGGX/ V,
        N,NOMGG/C,N,/C,N,/C,N,/C,Y,COUPMASS/C,Y,CPBAR/C,Y,CPROD/C,Y,
        CPQUAD1/C,Y,CPQUAD2/C,Y,CPTRIA1/C,Y,CPTRIA2/ C,Y,CPTUBE/C,Y,
        CPQDPLT/C,Y,CPTRPLT/C,Y,CPTRBSC/V,Y,DPMN $
```

The default value for DPMN, which is set in XMPLBD, is zero (0). This means if the PARAM card for DPMN is omitted, $[a_{ee}]$ will make no contribution in equation (14).

Subroutine SDR1A was modified to calculate the real part of $\{\phi\}_n e^{(\alpha + i\omega)t}$ where $\{\phi\}_n$ is the complex eigenvector generated by the module CEAD. To avoid entering the modified section of code each time SDR1A is called, a new parameter, IFLUT, was added to the DMAP calling sequence for module SDR1. The contents of IFLUT are passed through blank common from SDR1 to SDR1A. The default value for IFLUT, which is set in XMPLBD, is zero (0). When IFLUT = 0, the new code in SDR1A will not be executed. To set IFLUT = 1 and execute the new code in SDR1A, the following calling sequence for the SDR1 module is used:

```
SDR1    USET,PHIA,,,GØ,GM,KFS/PHIG,, BQG/1/*REIG*/1 $
```

The underlined parameter sets IFLUT to 1.

Once the changes were made to DBAR, KBAR, SDR1A, and XMPLBD, they were compiled and replaced the old DBAR, KBAR, SDR1A, and XMPLBD in the NASTRAN object library. Link 1, Link 3, Link 12, and Link 13 were relinked, creating a new executable NASTRAN. Although this procedure was done on a CDC computer, similar procedures will produce similar results on the IBM and UNIVAC computers.

To use this capability in NASTRAN, the DMAP sequence shown in Appendix A must be used. This sequence uses many of the new DMAP convenience features in Level 16 of NASTRAN. One of the features allows the REPT module to have a variable parameter. The variable parameter NLØØP is used for REPT in this DMAP sequence. NLØØP is input on a PARAM card in the BULK DATA deck. It sets the maximum number of iterations of the inner loop shown in figure 2. The only other input required to use this capability is the addition of another PARAM card in the BULK DATA deck. The parameter AMP corresponds to c and is used to specify the amplitude of vibration of this structure. This capability was added to an in-house version only and is not available in any standard NASTRAN level.

RESULTS AND DISCUSSION

The large deflection panel flutter analysis developed for use with NASTRAN has been applied to various panels. A typical BULK DATA deck for a simply supported panel at $\frac{c}{h} = 0.6$ and $\lambda = 600.0$ is given in Appendix B.

Convergence Study

Numerical results for the first two eigenvalues at $\lambda = 0$ and for the coalescence for a simply supported panel and a clamped panel are shown in Table 1. The exact results for eigenvalue coalescence are from reference 20. It is seen that an excellent approximation to the exact results is obtained with only eight elements.

The influence of large deflections on in-vacuo frequencies for a simply supported panel is given in Table 2. Analytical solutions using three different approaches from reference 21 are also given. Comparison of the NASTRAN results with the reference 21 methods show that the eight-element approximation gives very good results. Therefore, eight elements were used in modeling the panels in all the flutter results presented.

Simply Supported Panel and Effect of Aerodynamic Damping

Plots of the eigenvalues verses dynamic pressure for a simply supported panel at two different panel amplitudes $\frac{c}{h} = 0.0$ (linear theory) and 0.6, are shown in figure 4. The complete panel behavior is characterized by plotting the $(\alpha + i\omega)$ variation with increasing dynamic pressure λ, using equation (30) and figure 4, as shown in figure 5. For the case of negligible aerodynamic damping, $g_A \to 0$, instability does not set in until after the two undamped natural frequencies have merged. If some damping is present, the instability sets in at a somewhat higher value as indicated in figure 5. This occurs when $\alpha = 0$ in equation (30). By routine algebraic manipulation, this instability occurs at the value of λ when

$$g_A = \frac{\kappa_I}{\sqrt{\kappa_R}} \tag{32}$$

and the corresponding limit cycle frequency is

$$\frac{\omega}{\omega_o} = \sqrt{\kappa_R} \tag{33}$$

However, as discussed earlier, this instability is not catastrophic. The panel response does not grow indefinitely, but rather a limit cycle oscillation is developed with increasing amplitude as λ increases.

Boundary Support Effect

In figure 6, the panel amplitude of the limit cycle oscillation is given as a function of λ for various panel edge restraints. The most interesting result is that the limit cycle motions are different for hinged-clamped and clamped-hinged panels. This occurs because the aerodynamic matrices are different for the two support conditions, which leads to different deflection shapes for the panels as well as different geometrical stiffness matrices.

Effect of In-Plane Loading

Panel amplitude versus λ for several applied in-plane forces acting on a simply supported panel is shown in figure 7. The classical Euler buckling load for simply supported panels is $N_{cr} = -\pi^2 D/a^2$. The total membrane force is composed of the applied in-plane load N_{xo} and the membrane force N_x induced by large deflections of the panel. Figure 7 shows that the applied compressive in-plane force reduces the critical dynamic pressure. However, as the dynamic pressure is increased the panel amplitude increases, which induces tensile in-plane forces that counteract the applied compressive forces. This process continues until a flutter dynamic pressure is reached which corresponds to a given limit cycle amplitude.

CONCLUDING REMARKS

A large amplitude supersonic panel flutter capability has been developed for use with NASTRAN Level 16.0 by means of DMAP sequences and modifications to the code. An aerodynamic matrix for a two-dimensional plate element has been developed for NASTRAN by modifying subroutine KBAR. The iteration process has been implemented in NASTRAN through PARAM NLØØP in bulk data deck, modifications in subroutines DBAR, KBAR, and SDR1AZZ, and the DMAP sequences. Examples which include effects of aerodynamic damping, applied inplane forces and various support conditions have demonstrated the versatility of the method.

APPENDIX A

DMAP SEQUENCES

```
ID          NLPF,TWOD
APP         DMAP
BEGIN       $
XDMAP       GO,ERR=2,LIST.
FILE        LAMA=APPEND/PHIA=APPEND $
GP1         GEOM1,GEOM2,/GPL,EQEXIN,GPDT,CSTM,BGPDT,SIL/V,N,LUSET/  V,N,
            NOGPDT $
SAVE        LUSET $
CHKPNT      GPL,EQEXIN,GPDT,CSTM,BGPDT,SIL $
GP2         GEOM2,EQEXIN/ECT $
CHKPNT      ECT $
PARAML      PCDB//C,N,PRES/C,N,/C,N,/C,N,/V,N,NOPCDB $
PURGE       PLTSETX,PLTPAR,GPSETS,ELSETS/NOPCDB $
COND        P1,NOPCDB $
PLTSET      PCDB,EQEXIN,ECT/PLTSETX,PLTPAR,GPSETS,ELSETS/V,N,NSIL/   V,N,
            JUMPPLOT=-1 $
SAVE        NSIL,JUMPPLOT $
PRTMSG      PLTSETX// $
PARAM       //C,N,MPY/V,N,PLTFLG/C,N,1/C,N,1 $
PARAM       //C,N,MPY/V,N,PFILE/C,N,0/C,N,0 $
COND        P1,JUMPPLOT $
PLOT        PLTPAR,GPSETS,ELSETS,CASECC,BGPDT,EQEXIN,SIL,,,,/PLOTX1/   V,N,
            NSIL/V,N,LUSET/V,N,JUMPPLOT/V,N,PLTFLG/V,N,PFILE $
SAVE        JUMPPLOT,PLTFLG,PFILE $
PRTMSG      PLOTX1// $
LABEL       P1 $
CHKPNT      PLTPAR,GPSETS,ELSETS $
GP3         GEOM3,EQEXIN,GEOM2/SLT,GPTT/V,N,NOGRAV $
CHKPNT      SLT,GPTT $
TA1         ECT,EPT,BGPDT,SIL,GPTT,CSTM/EST,GEI,GPECT,/V,N,LUSET/   V,N,
            NOSIMP/C,N,1/V,N,NOGENL/V,N,GENEL $
SAVE        NOSIMP,NOGENL,GENEL $
COND        ERROR1,NOSIMP $
PURGE       OGPST/GENEL $
CHKPNT      EST,GPECT,GEI,OGPST $
PARAM       //C,N,ADD/V,N,NOKGGX/C,N,1/C,N,0 $
PARAM       //C,N,ADD/V,N,NOMGG/C,N,1/C,N,0 $
EMG         EST,CSTM,MPT,DIT,GEOM2,/KELM,KDICT,MELM,MDICT,,/V,N,NOKGGX/ V,
            N,NOMGG/C,N,/C,N,/C,N,/C,Y,COUPMASS/C,Y,CPBAR/C,Y,CPROD/C,Y,
            CPQUAD1/C,Y,CPQUAD2/C,Y,CPTRIA1/C,Y,CPTRIA2/  C,Y,CPTUBE/C,Y,
            CPQDPLT/C,Y,CPTRPLT/C,Y,CPTRBSC/V,Y,DPMN $
SAVE        NOKGGX,NOMGG $
CHKPNT      KELM,KDICT,MELM,MDICT $
COND        JMPKGG,NOKGGX $
```

```
EMA       GPECT,KDICT,KELM/KGGX,GPST $
CHKPNT    KGGX,GPST $
LABEL     JMPKGG $
COND      ERROR5,NOMGG $
EMA       GPECT,MDICT,MELM/MGG,/C,N,-1/C,Y,WTMASS=1.0 $
CHKPNT    MGG $
COND      LBL1,GRDPNT $
GPWG      BGPDT,CSTM,EQEXIN,MGG/OGPWG/V,Y,GRDPNT/C,Y,WTMASS $
OFP       OGPWG,,,,,// $
LABEL     LBL1 $
EQUIV     KGGX,KGG/NOGENL $
CHKPNT    KGG $
COND      LBL11,NOGENL $
SMA3      GEI,KGGX/KGG/V,N,LUSET/V,N,NOGENL/V,N,NOSIMP $
CHKPNT    KGG $
LABEL     LBL11 $
PARAM     //C,N,MPY/V,N,NSKIP/C,N,0/C,N,0 $
GP4       CASECC,GEOM4,EQEXIN,SIL,GPDT,BGPDT,CSTM/RG,YS,USET,ASET/V,N,
          LUSET/V,N,MPCF1/V,N,MPCF2/V,N,SINGLE/V,N,OMIT/V,N,REACT/V,N,
          NSKIP/V,N,REPEAT/V,N,NOSET/V,N,NOL/V,N,NOA/C,Y,SUBID $
SAVE      MPCF1,MPCF2,SINGLE,OMIT,REACT,NSKIP,REPEAT,NOSET,NOL,NOA $
COND      ERROR6,NOL $
PARAM     //C,N,AND/V,N,NOSR/V,N,SINGLE/V,N,REACT $
PURGE     GM/MPCF1/GO,KOO,LOO,PO,UOOV,RUOV/OMIT/PS,KFS,KSS/SINGLE/  QG/
          NOSR $
CHKPNT    GM,RG,GO,KOO,LOO,PO,UOOV,RUOV,YS,PS,KFS,KSS,USET,ASET,QG $
COND      LBL4D,REACT $
JUMP      ERROR2 $
LABEL     LBL4D $
COND      LBL4,GENEL $
GPSP      GPL,GPST,USET,SIL/OGPST/V,N,NOGPST $
SAVE      NOGPST $
COND      LBL4,NOGPST $
OFP       OGPST,,,,,// $
LABEL     LBL4 $
EQUIV     KGG,KNN/MPCF1/MGG,MNN/MPCF1 $
CHKPNT    KNN,MNN $
COND      LBL2,MPCF2 $
MCE1      USET,RG/GM $
CHKPNT    GM $
MCE2      USET,GM,KGG,MGG,, /KNN,MNN,, $
CHKPNT    KNN,MNN $
LABEL     LBL2 $
EQUIV     KNN,KFF/SINGLE/MNN,MFF/SINGLE $
```

```
CHKPNT   KFF,MFF $
COND     LBL3,SINGLE $
SCE1     USET,KNN,MNN,,/KFF,KFS,KSS,MFF,, $
CHKPNT   KFS,KSS,KFF,MFF $
LABEL    LBL3 $
EQUIV    KFF,KAA/OMIT/MFF,MAA/OMIT $
CHKPNT   KAA,MAA $
COND     LBL5,OMIT $
SMP1     USET,KFF,,,/GO,KAA,KOO,LOO,UOO,,,, $
CHKPNT   GO,KAA,KOO,LOO,UOO $
SMP2     USET,GO,MFF/MAA $
CHKPNT   MAA $
LABEL    LBL5 $
RBMG2    KAA/LLL $
CHKPNT   LLL $
SSG1     SLT,BGPDT,CSTM,SIL,EST,MPT,GPTT,EDT,MGG,CASECC,DIT/PG/ V,N,
         LUSET/C,N,1 $
CHKPNT   PG $
EQUIV    PG,PL/NOSET $
CHKPNT   PL $
COND     LBL10,NOSET $
SSG2     USET,GM,YS,KFS,GO,,PG/,PO,PS,PL $
CHKPNT   PO,PS,PL $
LABEL    LBL10 $
SSG3     LLL,KAA,PL,LOO,KOO,PO/ULV,UOOV,RULV,RUOV/V,N,OMIT/V,Y,IRES=-1/
         C,N,1/V,N,EPSI $
SAVE     EPSI $
CHKPNT   ULV,UOOV,RULV,RUOV $
COND     LBL9,IRES $
MATGPR   GPL,USET,SIL,RULV//C,N,L $
MATGPR   GPL,USET,SIL,RUOV//C,N,O $
LABEL    LBL9 $
SDR1     USET,PG,ULV,UOOV,YS,GO,GM,PS,KFS,KSS,/UGV,PGG,QG/C,N,1/C,N,
         BKLO $
CHKPNT   UGV,QG,PGG $
SDR2     CASECC,CSTM,MPT,DIT,EQEXIN,SIL,GPTT,EDT,BGPDT,,QG,UGV,EST,,PGG/
         OPG1,OQG1,OUGV1,OES1,OEF1,PUGV1/C,N,BKLO $
PARAM    //C,N,MPY/V,N,CARDNO/C,N,0/C,N,0 $
OFP      OUGV1,OPG1,OQG1,OEF1,OES1,//V,N,CARDNO $
SAVE     CARDNO $
COND     P2,JUMPPLOT $
PLOT     PLTPAR,GPSETS,ELSETS,CASECC,BGPDT,EQEXIN,SIL,PUGV1,,GPECT,OES1/
         PLOTX2/V,N,NSIL/V,N,LUSET/V,N,JUMPPLOT/V,N,PLTFLG/V,N,PFILE $
SAVE     PFILE $
```

```
PRTMSG   PLOTX2// $
LABEL    P2 $
TA1      ECT,EPT,BGPDT,SIL,GPDT,CSTM/X1,X2,ECPT,GPCT/V,N,LUSET/  V,N,
         NOSIMP/C,N,0/V,N,NOGENL/V,N,GENEL $
DSMG1    CASECC,GPTT,SIL,EDT,UGV,CSTM,MPT,ECPT,GPCT,DIT/KDGG/  V,N,
         DSCOSET $
SAVE     DSCOSET $
CHKPNT   KDGG $
DPD      DYNAMICS,GPL,SIL,USET/GPLD,SILD,USETD,,,,,,,EED,EQDYN/V,N,
         LUSET/V,N,LUSETD/V,N,NOTFL/V,N,NODLT/V,N,NOPSDL/V,N,NOFRL/  V,
         N,NONLFT/V,N,NOTRL/V,N,NOEED/C,N,/V,N,NOUE $
SAVE     NOEED $
COND     ERROR3,NOEED $
CHKPNT   EED $
PARAM    //C,N,MPY/V,N,NEIGV/C,N,1/C,N,-1 $
LABEL    NLVIB $
ADD      KAA,/KTT/C,N,(-1.0,0.0)/C,N,(0.0,0.0) $
CHKPNT   KTT $
CEAD     KTT,,MAA,EED,CASECC/PHIA,LAMA,OEIGS/S,N,NEIGV $
OFP      OEIGS,LAMA//S,N,CARDNO $
COND     ERROR4,NEIGV $
SDR1     USET,,PHIA,,,GO,GM,,KFS/PHIG,,BQG/1/*REIG*/1 $
ADD      PHIG/PHIAMP/V,Y,AMP $
DSMG1    CASECC,,SIL,,PHIAMP,CSTM,MPT,ECPT,GPCT,DIT/KNGG/DSCOSET/1 $
CHKPNT   KNGG $
ADD5     KGG,KDGG,KNGG,, /KSGG $
CHKPNT   KSGG $
EQUIV    KSGG,KSNN/MPCF2/MGG,MSNN/MPCF2 $
CHKPNT   KSNN,MSNN $
COND     LBL2S,MPCF2 $
MCE2     USET,GM,KSGG,MGG/KSNN,MSNN $
CHKPNT   KSNN,MSNN $
LABEL    LBL2S $
EQUIV    KSNN,KSFF/SINGLE/MSNN,MSFF/SINGLE $
CHKPNT   KSFF,MSFF $
COND     LBL3S,SINGLE $
SCE1     USET,KSNN,MSNN/KSFF,KSFS,,MSFF $
CHKPNT   KSFF,KSFS,MSFF $
LABEL    LBL3S $
EQUIV    KSFF,KSAA/OMIT / MSFF,MSAA/OMIT $
CHKPNT   KSAA,MSAA $
COND     LBL5S,OMIT $
SMP1     USET,KSFF/GSO,KSAA,KSOO,LSOO,USOO $
SMP2     USET,GSO,MSFF/MSAA $
```

```
CHKPNT    KSAA,MSAA $
LABEL     LBL5S $
COPY      KSAA/KAA/IPARM=-1 $
COPY      MSAA/MAA/JPARM=-1 $
REPT      NLVIB,NLOOP $
SDR2      CASECC,CSTM,MPT,DIT,EQEXIN,SIL,,,BGPDT,LAMA,BQG,PHIG,EST,,/,
          OBQG1,OPHIG,OBES1,OBEF1,PPHIG/C,N,REIG $
OFP       OPHIG,OBQG1,OBEF1,OBES1,,//V,N,CARDNO $
SAVE      CARDNO $
COND      P3,JUMPPLOT $
PLOT      PLTPAR,GPSETS,ELSETS,CASECC,BGPDT,EQEXIN,SIL,,PPHIG,GPECT,
          OBES1/PLOTX3/V,N,NSIL/V,N,LUSET/V,N,JUMPPLOT/V,N,PLTFLG/V,N,
          PFILE $
SAVE      PFILE $
PRTMSG    PLOTX3// $
LABEL     P3 $
JUMP      FINIS $
LABEL     ERROR1 $
PRTPARM   //C,N,-1/C,N,NMDS $
LABEL     ERROR2 $
PRTPARM   //C,N,-2/C,N,NMDS $
LABEL     ERROR3 $
PRTPARM   //C,N,-3/C,N,NMDS $
LABEL     ERROR4 $
PRTPARM   //C,N,-4/C,N,NMDS $
LABEL     ERROR5 $
PRTPARM   //C,N,-5/C,N,NMDS $
LABEL     ERROR6 $
PRTPARM   //C,N,-6/C,N,NMDS $
LABEL     FINIS $
END       $
```

APPENDIX B

INPUT BULK DATA CARDS

```
$
$ GEOMETRY AND CONSTRAINTS
GRDSET                                                          246
GRID     1               0.0
GRID     2               0.125
GRID     3               0.25
GRID     4               0.375
GRID     5               0.50
GRID     6               0.625
GRID     7               0.75
GRID     8               0.875
GRID     9               1.0
GRID     20              0.0      10.0                         123456
SPC      1       1       13       0.0      9        3          0.0
$
$ STRUCTURAL AND AERODYNAMIC ELEMENTS
BAROR            15                        20                          2
CBAR     1               1        2
CBAR     2               2        3
CBAR     3               3        4
CBAR     4               4        5
CBAR     5               5        6
CBAR     6               6        7
CBAR     7               7        8
CBAR     8               8        9
MATI*    25              1.0                                             *MT1
*MT1     0.4367901341
PARAM    COUPMASS1
$ DPMN  = 2.0*Q/(BETA)
$        WHERE Q = RHO*V**2/2.0, DYNAMIC PRESSURE
$                BETA = SQRT(MACH NO.**2 - 1.0)
PARAM    DPMN    600.0
PBAR*    15              25                2.289429        1.0           *PB1
*PB1     1.0
$
$ CONTROL DATA
EIGC     1       INV     MAX                                             +INV1
+INV1    41.0    -12.0   41.0     -14.0    1.00     1       1
$ AMP  = AMPLITUDE/SQRT(I/A) = SQRT(12.0)*C/H = SQRT(12.0)*0.6 = 2.078461
$        WHERE I = AREA MOMENT OF INERTIA
$              A = AREA
PARAM    AMP     2.0784610.0
PARAM    NLOOP   3
$
$ APPLIED INPLANE LOADING
FORCE*   1               9                                 9.86960440109 +FCE
+FCE     -1.0    0.0     0.0
ENDDATA
```

REFERENCES

1. Fung, Y. C.: On Two-Dimensional Panel Flutter, J. Aeronautical Sciences. Vol. 25, 1958, pp. 145-160.

2. Dugundji, J.: Theoretical Considerations of Panel Flutter at High Supersonic Mach Numbers. AIAA J., Vol. 4, 1966, pp. 1257-1266.

3. Dowell, E. H.: A Review of the Aeroelastic Stability of Plates and Shells. AIAA J., vol. 8, 1970, pp. 385-399.

4. Dowell, E. H.: Nonlinear Oscillations of a Fluttering Plate. AIAA J. Vol. 4, 1966, pp. 1267-1275.

5. Dowell, E. H. : Nonlinear Oscillations of a Fluttering Plate II. AIAA J., Vol. 5, 1967, pp. 1856-1862.

6. Eastep, F. E. and McIntosh, S. C.: The Analysis of Nonlinear Panel Flutter and Response under Random Excitation or Nonlinear Aerodynamic Loading. Proceedings AIAA/ASME 11th SDM Conference, Denver, Colo., April 1972, pp. 36-47.

7. Kuo, C. C., Morino, L. and Dugundji, J.: Perturbation and Harmonic Balance Methods for Nonlinear Panel Flutter. AIAA J., Vol 10, 1972, pp. 1479-1484.

8. Morino, L.: A Perturbation Method for Treating Nonlinear Panel Flutter Problems. AIAA J., Vol. 7, 1969, pp. 405-410.

9. Olson, M.S.: Finite Elements Applied to Panel Flutter. AIAA J. Vol. 5, 1967, pp. 2267-2270.

10. Kariappa and Somashekar, B.R.: Application of Matrix Displacement Methods in the Study of Panel Flutter. AIAA J., Vol. 7, 1969, pp. 50-53.

11. Olson, M.S.: Some Flutter Solutions Using Finite Elements. AIAA J., Vol. 8, 1970, pp. 747-752.

12. Kariappa, Somashekar, B. R. and Shah, C. G.: Discrete Element Approach to Flutter of Skew Panels with In-plane Forces under Yawed Supersonic Flow. AIAA J., Vol. 8, 1970, pp. 2017-2022.

13. Sander G., Bon C. and Geradin, M.: Finite Element Analysis of Supersonic Panel Flutter. Int. J. Numerical Methods in Engr., Vol. 7, 1973, pp. 379-394.

14. Rossettos, J. N. and Tong, P.: Finite-Element Analysis of Vibration and Flutter of Cantilever Anisotropic Plates. ASME Paper 74-WA/APM-15.

15. Bismarck-Nasr, M. N.: Finite Element Method Applied to the Supersonic Flutter of Circular Cylindrical Shells. Int. J. Numerical Methods in Engr., Vol. 10, 1976, pp. 423-435.

16. Mei, C.: Finite Element Displacement Method for Large Amplitude Free Flexural Vibrations of Beams and Plates. Int. J. Comp. & Stru., Vol. 3, 1973, pp. 163-174.

17. Mei, C. and Rogers, J. L., Jr.: NASTRAN Nonlinear Vibration Analysis of Beam and Frame Structures. NASA TM X-3278, 1975, pp. 259-284.

18. Rao, G. V., Raju, I. S. and Raju, K. K.: Nonlinear Vibrations of Beams Considering Shear Deformation and Rotatory Inertia. AIAA J., Vol. 14, 1976, pp. 685-687.

19. Bergan, P. G. and Clough, R. W.: Convergence Criteria for Iterative Processes. AIAA J., Vol. 10, 1972, pp. 1107-1108.

20. Olson, M. D.: On Applying Finite Elements to Panel Flutter. Aero. Rept. LR-476, National Research Council, Ottawa, Canada, March 1967.

21. Ray, J. D. and Bert, C. W.: Nonlinear Vibrations of a Beam with Pinned Ends. Trans. ASME, J. Engr. for Industry, Vol. 91, 1969, pp. 997-1004.

TABLE 1. IN VACUO EIGENVALUES AND COALESCENCE RESULTS FOR SIMPLY SUPPORTED AND CLAMPED PANELS

Simply Supported Panel

Number of Elements	In Vacuo κ_1	In Vacuo κ_2	Coalescence λ_{cr}	Coalescence κ_{cr}
2	98.1795	1920.00	398.536	1206.32
4	97.4597	1570.87	342.347	1043.47
8	97.4123	1559.35	343.280	1051.22
Exact (ref. 20)	97.4091	1558.55	343.3564	1051.797

Clamped Panel

Number of Elements	In Vacuo κ_1	In Vacuo κ_2	Coalescence λ_{cr}	Coalescence κ_{cr}
2	516.923	6720.00	922.388	3618.46
4	501.894	3874.23	636.437	2721.38
8	500.648	3808.34	636.586	2740.16
Exact (ref. 20)	500.564	3803.54	636.5691	2741.360

TABLE 2. EFFECT OF AMPLITUDE RATIO ON IN-VACUO FREQUENCY RATIOS $(\omega/\omega_o)_n$ FOR SIMPLY SUPPORTED PANEL

Amplitude $(\frac{c}{h})$	Mode n	Number of Elements 4	8	12	Theory (ref. 21) Assumed Space Mode	Assumed Time Mode	Galerkin
0.0	1	1.000	1.000	1.000	1.000	1.000	1.000
	2	1.004	1.000	1.000	1.000	-	-
0.2	1	1.038	1.039	1.040	1.056	1.032	1.048
	2	1.030	1.038	1.039	1.056	-	-
0.4	1	1.141	1.147	1.148	1.206	1.124	1.181
	2	1.106	1.141	1.146	1.206	-	-
0.6	1	1.292	1.304	1.306	1.411	1.262	1.375
	2	1.221	1.292	1.301	1.411	-	-
0.8	1	1.471	1.489	1.492	1.647	1.434	1.607
	2	1.367	1.471	1.484	1.647	-	-
1.0	1	1.667	1.690	1.693	1.902	1.627	1.863
	2	1.534	1.667	1.685	1.902	-	-
1.2	1	1.869	1.902	1.906	2.167	1.837	2.136
	2	1.716	1.870	1.895	2.167	-	-



Figure 1. Panel geometry.



Figure 2. Simplified flow diagram for large deflection panel flutter analysis.



Figure 3. Convergence characteristics.



Figure 4. Variation of eigenvalues with dynamic pressure for simply supported panel (N_{xo} = 0).



Figure 5. Typical plots of panel behavior and effect of aerodynamic damping (simply supported panel, $\frac{c}{h}$ = 0.6 and N_{xo} = 0).



Figure 6. Limit cycle amplitude versus dynamic pressure for panels with various support conditions ($N_{xo} = g_A = 0$).



Figure 7. Limit cycle amplitude versus dynamic pressure for simply supported panel under different in-plane forces ($g_A = 0$).